

SIMULAÇÕES



INFLAÇÃO
BOLSA DE VALORES
INTELIGÊNCIA ARTIFICIAL
SIMULAÇÕES DE VÔO
CONTROLES DE VÔO
JOGOS E DIVERTIMENTOS




McGRAW-HILL

OSCAR JULIO BURD

Simulações no MSX 

Simulações no MSX



Oscar Júlio Burd

McGraw-Hill
São Paulo
Rua Tabapuã, 1.105, Itaim-Bibi
CEP 04533
(011) 881-8604 e (011) 881-8528

Rio de Janeiro • Lisboa • Porto • Bogotá • Buenos Aires • Guatemala • Madrid • México • New York • Panamá •
San Juan • Santiago

Auckland • Hamburg • Kuala Lumpur • London • Milan • Montreal • New Delhi • Paris • Singapore • Sydney •
Tokyo • Toronto

Simulações no MSX

Copyright © 1987 da Editora McGraw-Hill, Ltda.

Todos os direitos para a língua portuguesa reservados pela Editora McGraw-Hill, Ltda.

Nenhuma parte desta publicação poderá ser reproduzida, guardada pelo sistema "retrieval" ou transmitida de qualquer modo ou por qualquer outro meio, seja este eletrônico, mecânico, dc fotocópia, de gravação, ou outros, sem prévia autorização, por escrito, da Editora.

Editor: Milton Mira de Assumpção Filho

Coordenadora de Revisão: Daisy Pereira Daniel

Supervisor de Produção: Neder Roberto O. Campos

Capa: Layout: Cyro Giordano

Arte final: Ademir Aparecido Alves

**Dados de Catalogação na Publicação (CIP) Internacional
(Câmara Brasileira do Livro, SP, Brasil)**

Burd, Oscar Julio.

B969s Simulações no MSX / Oscar Júlio Burd. -- São Paulo: McGraw-Hill, 1987

1. MSX (Computadores) – Métodos de simulação
2. Microcomputadores – Métodos de simulação I. Título.

87-0583

CDD-001.64040724

Índices para catálogo sistemático:

1. MSX : Microcomputadores : Simulação : Processamento de dados 001.64040724
2. Simulação ; Microcomputadores : Processamento de dados 001.64040724



1 - EVITANDO FALHAS	1
2 - ALGUMAS PALAVRAS SOBRE SIMULAÇÕES	2
3 - XADREZ	4
4 - LISTAGEM DO XADREZ	7
5 - O PRESIDENTE DA REPÚBLICA	17
6 - LISTAGEM DE O PRESIDENTE DA REPÚBLICA	22
7 - INTELIGÊNCIA ARTIFICIAL-I	31
8 - LISTAGEM DO LIGA QUATRO	37
9 - INTELIGÊNCIA ARTIFICIAL-II	46
10 - LISTAGEM DO MSX-THELLO	51
11 - BOLSA DE VALORES	56
12 - LISTAGEM DA BOLSA DE VALORES	68
13 - E AGORA... VOAR!!	75
14 - INTRODUÇÃO AOS CONTROLES DE VÔO	81
15 - LISTAGEM DO SIMULADOR DE VÔO	86



EVITANDO FALHAS

Todas as simulações deste livro foram escritas e testadas em um microcomputador MSX nacional e as listagens impressas a partir deste micro.

Todos os programas foram escritos em MSX-BASIC elementar, de modo a evitar dificuldades de compreensão por parte dos leitores.

Eventuais erros que possam interromper as simulações terão sua origem em possíveis erros de digitação cometidos pelos usuários.

Caso ocorra um erro de **FALTA DE MEMÓRIA** aconselho que desligue e ligue novamente seu MSX para, então, executar o programa. Esse tipo de erro tem sua origem na execução anterior de outro programa que possa ter alterado os vetores do seu micro.

Aconselho o máximo de atenção quando digitar os programas contidos neste livro.

A cada quinze minutos de digitação grave o programa. Faça uma cópia do programa em disco ou K-7 ao finalizar a digitação.



ALGUMAS PALAVRAS SOBRE SIMULAÇÕES

Seja bem-vindo ao mundo das simulações!

Neste mundo quase tudo é possível; a final de contas as simulações lhe dão a oportunidade de transportar o mundo "real" para dentro do seu MSX!

Que tal ser o Presidente da República? Gostaria de pilotar um avião e ter todos os instrumentos "realmente" funcionando à sua frente? Talvez você prefira enfrentar o seu computador em alguns jogos que requerem um bom tanto de inteligência e destreza?

Pois bem, independente da sua resposta, este livro colocará ao seu alcance todas as situações acima e mais: pouco a pouco você entenderá como funcionam as simulações e de que maneira são projetadas. Em outras palavras, se você conhece um pouco de MSX-BASIC, será capaz de projetar e programar suas próprias simulações!!

O estudo das listagens deste livro e do modo como são estruturados os programas lhe dará os elementos necessários ao seu desenvolvimento no mundo das simulações.

As simulações por computador são utilizadas em três situações básicas: quando a realidade é muito cara, quando a realidade é muito perigosa ou quando a realidade é muito complexa.

As primeiras horas de vôo de pilotos de grandes aviões são passadas em simuladores de vôo controlados por computadores; os próprios astronautas da NASA passam boa parte de seus treinamentos em simuladores.

Existem sistemas que simulam a reação de seres humanos a certas substâncias químicas, como se elas estivessem dentro de seus organismos.

Recentemente, no Brasil, algumas auto-escolas passaram a adotar simuladores de veículos para seus alunos novatos. Sem dúvida essa medida implica em grande economia por parte da auto-escola, não apenas em termos materiais mas também em termos temporais (podem atender a um número maior de alunos com um bom nível de ensino).

Ainda no Brasil, uma grande cadeia de supermercados está estudando a viabilidade de utilizar computadores no treinamento de seus caixas: os equipamentos simulariam as situações do dia-a-dia de um caixa, inclusive os horários de maior movimento.

O treinamento dos operadores de grandes usinas nucleares também é feito com a utilização de simulações controladas por computadores.

TIPOS DE SIMULAÇÕES

Simulações em tempo real são aqueles programas que fazem com que os fatos ocorram no mesmo tempo que gastam no mundo real; um segundo de simulação deve corresponder a um segundo no nosso mundo real.

O programa **SIMULADOR DE VÔO** pode ser transformado em uma simulação em tempo real (veja os comentários na listagem).

Já o programa **BOLSA-DE VALORES** não ocorre em tempo real, pois a bolsa de valores “espera” até que você tome suas decisões.

O programa **BOLSA DE VALORES** é uma simulação discreta, pois ele espera que você digite uma tecla para prosseguir operando.

A NOSSA JORNADA

Começaremos simulando o final de uma partida de xadrez; você verá como é impossível vencer o seu micro!

Em seguida passaremos ao mundo da política, onde você é o Presidente da República do Brasil! Em pouco tempo descobrirá que a presidência não é um cargo dos mais agradáveis...

Prosseguindo, vamos simular um pouco de inteligência artificial voltada para jogos de tabuleiro, onde você conhecerá poderosos programas de jogos (o seu MSX simulará o seu adversário). Quem vencerá? Você ou o seu MSX?

Entraremos, então, no mundo da bolsa de valores, onde você é um grande acionista com grandes objetivos financeiros a atingir.

E, para finalizar, você entrará em um avião, ocupará o lugar do piloto e levantará vôo (após algumas tentativas, é claro!).

Descobrirá que não é simples a tarefa de controlar cinco ou mais fatores que se modificam a cada segundo.

Você tem um campo maravilhoso à sua frente, todo um universo a ser descoberto.

Vá lá!!



XADREZ

Existem programas de computadores que simulam muito bem jogadores de xadrez e seus adversários.

Mas qual foi a primeira máquina a simular tal jogo? Quem foi o seu inventor?

A resposta é a seguinte: em 1890, na Espanha, um cientista chamado Torres Quevedo criou uma pequena máquina que era capaz de jogar uma final de xadrez entre um rei e uma torre e um rei. A máquina jogava com a torre e o rei e era sempre capaz de dar xeque-mate (vencer).

A máquina podia fazer seis movimentos diferentes, que são indicados nesta simulação.

Lembre-se de que a torre desloca-se horizontal ou verticalmente, na linha em que está posicionada, qualquer número de casas (para maiores detalhes consulte alguém que jogue xadrez) e o rei anda de casa em casa em qualquer direção.

Você tem apenas um rei (símbolo "r" nas telas abaixo) e o seu MSX tem um rei ("R") e uma torre ("T").

As posições iniciais são sorteadas pelo micro:

Você usou o movimento: 0

ABCDEFGHI

8	8
7	T. 7
6	6
5	...R....	5
4	4
3	3
2	2
1	.r.....	1

ABCDEFGHI

Digite sua jogada? C2

Movimentei o meu rei (r) para a coluna C linha 2 (jogada C2). O micro deslocou a torre (T) para G6:

Voce usou o movimento: 1

ABCDEFGHI

8	8
7	7
6	T. 6
5	...R....	5
4	4
3	3
2	...r....	2
1	1

ABCDEFGHI

Digite sua jogada? D3

Joguei em D3; ele colocou a Torre (T) em G5:

Voce usou o movimento: 1

ABCDEFGHI

8	8
7	7
6	6
5	...R..T.	5
4	4
3	...r....	3
2	2
1	1

ABCDEFGHI

Digite sua jogada? C4

Por descuido coloquei o meu rei em C4, infelizmente, ficando a mercê do rei adversário (R):

Voce usou o movimento: 1

ABCDEFGHI

8	8
7	7
6	6
5	...R..T.	5
4	...r....	4
3	3
2	2
1	1

ABCDEFGHI

G A N H E I !!!

Note que você verá, na tela do seu MSX, os caracteres com desenhos diferentes (verá dois reis e uma torre), pois eles são redesenhadados na simulação.

A listagem do XADREZ está no próximo capítulo.



LISTAGEM DO XADREZ

Digite "F" para encerrar a simulação.

Caso você queira, pode modificar o formato das suas peças e das do computador: o desenho da sua peça é definido nas linhas de 2610 a 2680 e o desenho das peças do MSX é definido nas linhas de 2250 a 2320 e de 2370 a 2440 (rei e torre).

Note que aos uns correspondem os pontos cheios nas figuras das peças e aos zeros correspondem os espaços vazios.

Após executar esta simulação os formatos das letras "r", "R" e "T" estarão alterados; para fazer com que voltem ao normal digite:

SCREEN 0 ←

Esta simulação só lê os comandos em letras maiúsculas; antes de iniciá-la coloque o teclado na modalidade de maiúsculas!

1000 : *****

1010 : * XADREZ *

1020 : *****

1030 :

1040 :

1050 CLEAR&H1000: GOSUB 2050: REM INICIALIZACAO

1060 GOTO 1680

```
1070 * ****
1080 * LIMITE SUPERIOR *
1090 * ****
1100 IF INT(T/10) = INT(RB/10)+1 THEN RETURN
1110 IF T-10 = RN OR T-21 = RB OR T-19 = RB OR T-9 = R
B THEN GOSUB 1170: GOTO 1100
1120 T(T) = 46: T = T-10: T(T) = ASC("T"): SI = 1: MOV
I = 1
1130 GOSUB 1740: GOTO 1100
1140 * ****
1150 *MOVTO. HORIZONTAL*
1160 * ****
1170 CT = T-INT(T/10)*10
1180 DIR = 1: IF CT>4 THEN LET DIR = -1
1190 IF T+DIR = RN THEN LET DIR = -DIR
1200 T(T) = 46: T = T+DIR
1210 FOR A = 1 TO 6
1220 W = T-INT(T/10)*10: IF W = 1 OR W = 8 OR T = RN TH
EN LET A = 7: GOTO 1240
1230 T = T+DIR
1240 NEXT A
1250 T(T) = ASC("T"): MOVI = 2
1260 GOSUB 1740: RETURN
```

```
1270 * *****
1280 * *PREPARA POSICAO*
1290 * *****
1300 GOSUB 1100
1310 IF T-11 = RB OR T-9 = RB THEN GOSUB 1170: GOTO 13
00
1320 IF INT(RN/10) = INT(RB/10)+2 THEN RETURN
1330 T(RN) = 46: RN = RN-10: MOVI = 3: T(RN) = ASC("R")
1340 GOSUB 1740: GOTO 1300
1350 * *****
1360 *      POSICAO      *
1370 * *****
1380 SI = 0: GOSUB 1100: IF SI THEN GOTO 1700
1390 IF T-11 = RB OR T-9 = RB THEN GOSUB 1170: GOTO 17
00
1400 IF ABS(RN-20-RB) <= 1 THEN RETURN
1410 T(RN) = 46: RN = RN+1
1420 IF RN-20-RB > 0 THEN LET RN = RN-2
1430 T(RN) = ASC("R"): MOVI = 4
1440 GOSUB 1740: GOTO 1380
1450 * *****
1460 * *PREPARA MARCHA A RE* *
1470 * *****
```

```
1480 SI = 0: GOSUB 1100: IF SI THEN GOTO 1700
1490 IF T-11 = RB OR T-9 = RB THEN GOSUB 1170: GOTO 17
00
1500 CRB = RB-INT(RB/10)*10
1510 IF CRB = 1 OR CRB = 8 OR RB = RN- 20 THEN RETURN
1520 T(RN) = 46: RN = RN+1
1530 IF RN-INT(RN/10)*10-RB+INT(RB/10)*10 > 0 THEN LET
RN = RN-2
1540 T(RN) = ASC("R"): MOVI = 5
1550 GOSUB 1740: GOTO 1480
1560 * *****
1570 * *MARCHA A RE* *
1580 * *****
1590 SI = 0: GOSUB 1100: IF SI THEN GOTO 1700
1600 IF T-11 = RB OR T-9 = RB THEN GOSUB 1170: GOTO 17
00
1610 T(T) = 46: T = T-10: T(T) = ASC("T")
1620 GOSUB 1740: T(RN) = 46: RN = RB+20
1630 T(RN) = ASC("R"): MOVI = 6
1640 GOSUB 1740: GOTO 1720
1650 * *****
1660 * * LOOP PRINCIPAL *
1670 * *****
```

1680 GOSUB 1750: REM PEGA SEU MOVIMENTO
1690 GOSUB 1070:REM MOVIMENTO 1
1700 GOSUB 1280:REM MOVIMENTO 3
1710 GOSUB 1360:REM MOVIMENTO 4
1720 GOSUB 1460:REM MOVIMENTO 5
1730 GOSUB 1590:REM MOVIMENTO 6
1740 "
1750 " *****
1760 " * LE SUA JOGADA *
1770 " *****
1780 GOSUB 1940
1790 PLAY"A": LOCATE 0,21: INPUT "Digite sua jogada";G
\$
1800 IF G\$ = "f" OR G\$ = "F" THEN END
1810 IF LEN(G\$) <> 2 OR LEFT\$(G\$,1)<"A" OR LEFT\$(G\$,1)
>"H" THEN GOTO 1790
1820 X = VAL(MID\$(G\$,2,1)): IF X<1 OR X>8 THEN GOTO 17
90
1830 IF ABS(X-INT(RB/10)) > 1 THEN GOTO 1790
1840 IF ABS(RB-INT(RB/10)*10-ASC(LEFT\$(G\$,1))+64) = 0
AND ABS(X-INT(RB/10)) = 0 THEN GOTO 1790
1850 IF ABS(RB-INT(RB/10)*10-ASC(LEFT\$(G\$,1))+64) > 1
THEN GOTO 1790

```
1860 T(RB) = 46: RB = X*10+ASC(LEFT$(G$, 1))-64
1870 T(RB) = ASC("r"): GOSUB 1940: MOVI = 0
1880 REM VERIFICA SE PODE COMER O REI
1890 FRB = INT(RB/10): CRB = RB-FRB*10
1900 IF INT (T/10) = FRB OR T-INT(T/10)*10 = CRB THEN
      GOTO 2030
1910 IF ABS(RN-RB)<1 OR RN = RB+10 OR RN = RB-10 OR RN
= RB+9 OR RN = RB -9 OR RN = RB+11 OR RN = RB-11 THEN
      GOTO 2030
1920 RETURN
1930 " ****"
1940 " *IMPRIME TABULEIRO*
1950 " ****"
1960 CLS: PRINT "Voce usou o movimento:"MOVI
1970 LOCATE 10,5:PRINT "ABCDEFGH": PRINT
1980 FOR A = 80 TO 10 STEP -10: PRINT TAB(7)A/10;
1990 FOR B = 1 TO 8
2000 PRINT CHR$(T(A+B));
2010 NEXT B: PRINT A/10
2020 NEXT A: PRINT: PRINT TAB(10)"ABCDEFGH": RETURN
2030 PRINT: PRINT "G A N H E I !!!": PRINT: STOP
```

```
2040 * ****
2050 * * INICIALIZACAO *
2060 * ****
2070 SCREEN 1: KEY OFF:GOSUB 2180
2080 DIM T(100): FOR A=1 TO 100: LET T(A) = 46: NEXT A
2090 F = INT (RND(-TIME)*5)+1: C = INT (RND(-TIME)*8)+1
2100 RB = F*10+C: REM POSICAO REI BRANCO
2110 FBIS = F+2+INT(RND(-TIME)*6): IF FBIS>8 THEN LET FBIS=8
2120 RN = FBIS*10+INT(RND(-TIME)*7)+1: REM POSICAO DO REI PRETO
2130 FBIS = F+2+INT(RND(-TIME)*6): IF FBIS > 8 THEN LET FBIS = 8
2140 T = FBIS*10+INT(RND(-TIME)*8)+1: REM POSICAO DA TORRE PRETA
2150 IF T = RN THEN GOTO 2130
2160 T(T) = ASC("T"): T(RB) = ASC("r"): T(RN) = ASC("R")
2170 MOVI = 0
2180 * ****
2190 * * REDEFINE LETRAS *
2200 * ****
```

```
2210 RESTORE 2250: FOR W=0 TO 7
2220 READ A$
2230 VPOKE 656+W,VAL("&B"+A$)
2240 NEXT W
2250 DATA 01010100
2260 DATA 01111100
2270 DATA 01010100
2280 DATA 01111100
2290 DATA 01101100
2300 DATA 01111100
2310 DATA 00111000
2320 DATA 00010000
2330 RESTORE 2370: FOR W=0 TO 7
2340 READ A$
2350 VPOKE 672+W,VAL("&B"+A$)
2360 NEXT W
2370 DATA 11111100
2380 DATA 11111100
2390 DATA 01111100
2400 DATA 01111100
2410 DATA 01111100
2420 DATA 01111100
2430 DATA 01111100
```

```
2440 DATA 00000000
2450 RESTORE 2490: FOR W=0 TO 7
2460 READ A$
2470 VPOKE 368+W,VAL("&B"+A$)
2480 NEXT W
2490 DATA 00000000
2500 DATA 00000000
2510 DATA 00000000
2520 DATA 00110000
2530 DATA 00110000
2540 DATA 00000000
2550 DATA 00000000
2560 DATA 00000000
2570 RESTORE 2610: FOR W=0 TO 7
2580 READ A$
2590 VPOKE 912+W,VAL("&B"+A$)
2600 NEXT W
2610 DATA 01010101
2620 DATA 01000001
2630 DATA 01010101
2640 DATA 01000001
2650 DATA 01001001
2660 DATA 01000001
```

2670 DATA 01100011

2680 DATA 01110111

2690 RETURN



O PRESIDENTE DA REPÚBLICA

Passamos a simular, neste capítulo, nada mais, nada menos do que a economia de nosso país, a economia do Brasil.

Isso só é possível devido a existência de uma série de fórmulas matemáticas que representam, de forma muito simplificada, a economia de nossa nação.

Pois bem, a partir deste momento você passa a ocupar a presidência da república, assumindo também todas as responsabilidades inerentes a esse cargo.

Como Presidente você deve esforçar-se para conter os gastos públicos, equilibrar a inflação, manter reduzido o número de desempregados e, principalmente, manter a economia com um bom crescimento.

Acabará descobrindo que a vida de um Presidente nem sempre é tão boa como alguns pensam...

A simulação inicia com um informe sobre o estado geral da nação, sendo então solicitadas suas decisões sobre os gastos do Governo, o valor do salário mínimo e a permissão ou não da entrada de imigrantes no Brasil:

Presidente:Oscar
No poder ha' .25 anos.

----- SITUACAO DO PAIS -----

Populacao: 3000819
Desempregados: 300000 9 %
Sal.minimo:\$ 100 Inflacao: 5 %

Gastos no ultimo periodo:\$M 118
Custo do desemprego:\$M 0

Entradas por impostos:\$M 0
Caixa atual:\$M 0
Produto Nacional Bruto:\$M 440
Invest. publicos: \$M 236

Sr.Presidente Oscar:

Informe os gastos do governo \$M? 15

Presidente:Oscar
No poder ha' .25 anos.

----- SITUACAO DO PAIS -----

Populacao: 3000819
Desempregados: 300000 9 %
Sal.minimo:\$ 100 Inflacao: 5 %

Gastos no ultimo periodo:\$M 118
Custo do desemprego:\$M 0

Entradas por impostos:\$M 0
Caixa atual:\$M 0
Produto Nacional Bruto:\$M 440
Invest. publicos: \$M 236

Sr.Presidente Oscar:

e os gastos com salarios \$M? 100

Presidente: Oscar
No poder ha' .25, anos.

----- SITUACAO DO PAIS -----

Populacao: 3000819
Desempregados: 300000 9 %
Sal.minimo:\$ 100 Inflacao: 5 %

Gastos no ultimo periodo:\$M 118
Custo do desemprego:\$M 0

Entradas por impostos:\$M 0
Caixa atual:\$M 0
Produto Nacional Bruto:\$M 440
Invest. publicos: \$M 236

Sr.Presidente Oscar:

E' a favor dos imigrantes (S/N) ?
OK... N

Reduzi os gastos do governo para 15 milhões de cruzados, mantendo fixo o salário mínimo.

Abaixo, podemos perceber os resultados dessas medidas: redução do número de desempregados, pequeno aumento da inflação e aumento dos investimentos realizados pelo povo em obras públicas (isso significa que o povo está confiante na atual administração).

Vamos ao próximo período:

Presidente: Oscar
No poder ha' .5 anos.

----- SITUACAO DO PAIS -----

Populacao: 3001638
Desempregados: 217642 7 %
Sal.minimo:\$ 100 Inflacao: 7 %

Gastos no ultimo periodo:\$M 15
Custo do desemprego:\$M 21.7

Entradas por impostos:\$M 111.3
Caixa atual:\$M 74.5
Produto Nacional Bruto:\$M 529.3
Invest. publicos: \$M 414.3

Sr.Presidente Oscar:

Informe os gastos do governo \$M? 30

Aumentei os gastos para 30 milhões, o que fez, como podemos ver abaixo, com que a inflação e o desemprego aumentassem, além de diminuir o meu prestígio junto a população (os investimentos públicos caíram muito!):

Presidente:Oscar
No poder ha' .75 anos.

----- SITUACAO DO PAIS -----

Populacao: 3002457
Desempregados: 356275 11 %
Sal.minimo:\$ 100 Inflacao: 12 %

Gastos no ultimo periodo:\$M 30
Custo do desemprego:\$M 35.6

Entradas por impostos:\$M 105.8
Caixa atual:\$M 114.7
Produto Nacional Bruto:\$M 708.9
MUDANCA NO NIVEL DE VIDA 50 %
Invest. publicos: \$M 183.4

Sr.Presidente Oscar:
Informe os gastos do governo \$M? 60

Aumentei ainda mais os gastos (60 milhões) e veja o que ocorreu:

Presidente:Oscar
No poder ha' 1 anos.

----- SITUACAO DO PAIS -----

Populacao: 3003276
Desempregados: 199131 6 %
Sal.minimo:\$ 100 Inflacao: 6 %

Gastos no ultimo periodo:\$M 60
Custo do desemprego:\$M 19.9

Entradas por impostos:\$M 112.1
Caixa atual:\$M 146.9
Produto Nacional Bruto:\$M 523.8
MUDANCA NO NIVEL DE VIDA 16 %
Invest. publicos: \$M 729.1

Sr.Presidente Oscar:
e os gastos com salarios \$M? 200

Como a situação geral do país parecia ser favorável, aproveitei e aumentei o salário mínimo para Cz\$ 200.

Pensei que conquistaria maior simpatia junto ao povo, mas ocorreu um grande imprevisto:

A linha economica de sua administracao
levou o nosso pais a um inaceitavel
indice inflacionario.

A falta de eficacia na sua administra-
cao e' tal que foi solicitada a sua re-
nuncia...

O vice-presidente passa a ocupar o seu
cargo.

TECLE ALGO.

RELATORIO DE RENUNCIA:

Oscar foi presidente por 1.25 anos.

Durante o seu mandato a populacao
cresceu 3276 pessoas.

A taxa de desemprego passou a ser 17
%.

A taxa de inflacao era de 18.5 %.

O balanco passou a ser \$M 182.9 .

A listagem e os comentários referentes a O PRESIDENTE DA REPÚBLICA estão no capítulo seguinte.



LISTAGEM DE O PRESIDENTE DA REPÚBLICA

Esta é uma simulação que permite muitas alterações.

Caso deseje alterar os valores iniciais das variáveis, sugiro que dê uma olhada nas linhas de números 2380 a 2520, pois é lá que as variáveis recebem os valores iniciais.

Para modificar a impressão da tela e a entrada dos dados estude as linhas de números 1270 a 1610 e, depois, realize as alterações que tem em mente.

Se desejar novas mensagens quando da finalização da simulação, altere os textos contidos nas linhas de 2110 a 2320.

Finalmente, convém que você procure entender o que ocorre no LOOP PRINCIPAL (linhas de 1110 a 1230), pois é essa região do programa que “controla” toda a simulação.

Existe um grande número de comentários, nas linhas do programa, com o objetivo de facilitar a sua compreensão.

1000 " O P R E S I D E N T E

1010 "

1020 " D A

1030 "

1040 " R E P U B L I C A

1050 "

1060 '
1070 GOSUB 2350: REM INICIALIZACAO
1080 ' *****
1090 ' LOOP PRINCIPAL
1100 ' *****
1110 P = INT(P+(P*273/ML))
1120 GOSUB 1270: REM IMPRESSAO
1130 GOSUB 1650: REM CALCULOS
1140 ' AQUI E' VERIFICADO O FIM DA SIMULACAO
1150 GOSUB 1860: REM NIVEL DE VIDA
1160 GOSUB 2590: REM ACABOU?
1170 GOSUB 1940: REM NIVEL DE INFLACAO
1180 GOSUB 2590: REM ACABOU?
1190 GOSUB 2010: REM DESEMPREGO
1200 GOSUB 2590: REM ACABOU?
1210 GOSUB 1790: REM DEFICIT
1220 GOSUB 2590: REM ACABOU?
1230 GOTO 1110
1240 ' *****
1250 ' IMPRESSAO
1260 ' *****
1270 CLS
1280 PRINT "Presidente:";A\$

```
1290 PRINT "No poder ha' ";Y+Z/4;" anos."
1300 PRINT: "=====
1310 PRINT"----- SITUACAO DO PAIS -----"
1320 PRINT:"-----"
1330 PRINT "Populacao:";P
1340 PRINT "Desempregados:"INT(U) " "INT(100*U/P)"%"
1350 PRINT "Sal.minimo:$"INT(W0) " Inflacao:"INT(IP) "
%""
1360 PRINT:PRINT "Gastos no ultimo periodo:$M"INT(GE)
1370 PRINT "Custo do desemprego:$M"INT(10*GU)/10
1380 PRINT:PRINT "Entradas por impostos:$M"INT(GI*10)/
10
1390 PRINT "Caixa atual:$M"INT(BD*10)/10
1400 PRINT "Produto Nacional Bruto:$M"INT(GDP*10)/10
1410 IF Y+Z/4 > .5 THEN PRINT "MUDANCA NO NIVEL DE VID
A "INT((2*((RGDP/AGDP)*100)-100)/3) "%"
1420 PRINT "Invest. publicos: $M "INT(IV*10)/10
1430 PRINT"-----"
1440 PRINT: PRINT "Sr.Presidente "A$":"
1450 GOSUB 2610: INPUT "Informe os gastos do governo $M";GE
1460 GOSUB 2610
1470 INPUT "e os gastos com salarios $M";WN
```

```
1480 WN=WN+1E-03
1490 GOSUB 2610
1500 PRINT "E" a favor dos imigrantes (S/N) ?"
1510 X$ = INKEY$
1520 IF X$<>"S" AND X$<>"N" AND X$<>"s" AND X$<>"n" THEN
    GOTO 1510
1530 PRINT TAB(20)"OK... "X$
1540 FOR H=1 TO 500: NEXT H
1550 IF X$<>"S" AND X$<>"s" THEN RETURN
1560 GOSUB 2610
1570 PRINT "Permitira" a entrada de quantos imigrantes";
1580 INPUT M
1590 IF MK0 THEN GOTO 1580
1600 P = P+M
1610 RETURN
1620 *****
1630          CALCULOS
1640 *****
1650 CN = CN+(CN*IP/100)
1660 U = P*(GE+IV)/(CN*10)+P*(IP/1000)
1670 GU = U*WN/ML: REM CUSTO DO DESEMPREGO
1680 GI = (((P-U)*WN*.4)/ML): REM RENDA DOS IMPOSTOS
```

```
1690 BD = BD+GI-GU-GE: REM DEFICIT
1700 AGDP = AGDP*(1+(IP/100))
1710 GDP = GE+IV+((P-U)*WN/ML)
1720 RGDP = GDP*440/AGDP
1730 IP = ((GE+IV)/CN*.1+(WN/W0)/100)*100
1740 IV = (CN*67)/(IP*IP)
1750 W0 = WN
1760 Z = Z+1: IF Z>4 THEN LET Z=1: LET Y= Y+1
1770 RETURN
1780 > ****
1790 >      COMPROVA O DEFICIT
1800 > ****
1810 IF BD > -1000 THEN RETURN
1820 GAME = 1
1830 FLAG = 1
1840 RETURN
1850 > ****
1860 >      COMPROVA O NIVEL DE VIDA
1870 > ****
1880 IF Y<.75 THEN RETURN
1890 IF INT (((2*((RGDP/AGDP)*100)-100)/3) > -15 THEN R
ETURN
1900 GAME = 1
```

1910 FLAG = 2
1920 RETURN
1930 ? *****
1940 ? COMPROVA A INFLACAO
1950 ? *****
1960 IF IP<15 THEN RETURN
1970 GAME = 1
1980 FLAG = 3
1990 RETURN
2000 ? *****
2010 ? COMPROVA O DESEMPREGO
2020 ? *****
2030 IF INT(U*100/P)<15 THEN RETURN
2040 GAME = 1
2050 FLAG = 4
2060 RETURN
2070 ? *****
2080 ? FINALIZA A SIMULACAO
2090 ? *****
2100 CLS
2110 PRINT "Presidente"A\$":"
2120 PRINT:PRINT" A linha economica de sua administracao"
2130 PRINT "levou o nosso pais a um inaceitavel"

```
2140 IF FLAG = 1 THEN PRINT "deficit de balanca."
2150 IF FLAG = 2 THEN PRINT "nivel de vida."
2160 IF FLAG = 3 THEN PRINT "indice inflacionario."
2170 IF FLAG = 4 THEN PRINT "numero de desempregados"
2180 PRINT: PRINT " A falta de eficacia na sua adminis-
tra-"
2190 PRINT "cao e' tal que foi solicitada a sua re-"
2200 PRINT "nuncia...": PRINT
2210 FOR H=1 TO 1000:NEXT H
2220 PRINT " O vice-presidente passa a ocupar o seu"
2230 PRINT "cargo.": PRINT
2240 LOCATE 15,23: PRINT "TECLE ALGO."
2250 IF INKEY$ = "" THEN 2250
2260 CLS: PRINT " RELATORIO DE RENUNCIA": PRINT
2270 PRINT " "A$ " foi presidente por "Y+(Z*.25)" anos
.": PRINT
2280 PRINT " Durante o seu mandato a populacao": P
RINT
2290 PRINT "cresceu "P-3*ML" pessoas.": PRINT
2300 PRINT " A taxa de desemprego passou a ser ";: PRI
NT INT(U*1000/P/10) "%." : PRINT
2310 PRINT " A taxa de inflacao era de"INT(10*IP)/10"%
.": PRINT
```

```
2320 PRINT " O balanco passou a ser $M ";: PRINT INT(1
0*BD)/10 "."
2330 LOCATE 0,21: STOP
2340 * *****
2350 *      INICIALIZACAO
2360 * *****
2370 SCREEN0: KEY OFF
2380 ML = 1000000#
2390 P = 3*ML: REM POPULACAO
2400 U = P/10: REM DESEMPREGADOS
2410 IV = 236: REM INVERSAO
2420 GE = 118: REM GASTO DO GOVERNO
2430 GU = 0 : REM CUSTO DO DESEMPREGO
2440 GI = 0 : REM $ POR IMPOSTOS
2450 WN = 100: REM NOVOS SALARIOS
2460 WD = 100: REM ANTIGOS SALARIOS
2470 IP = 5 : REM % DE INFLACAO
2480 GDP = 440:REM PROD.NACION.BRUTO
2490 AGDP= 440:REM BASE DO P.NAC.BRUTO
2500 RGDP= 440:REM PNB REAL
2510 CN = 354: REM CONSTANTE ECONOMICA
2520 Z=1: GAME=0: FLAG=0
2530 Y=0:      REM ANOS NA PRESIDENCIA
```

```
2540 PRINT "DIGITE O SEU NOME, PRESIDENTE"  
2550 INPUT A$  
2560 BD = 0  
2570 RETURN  
2580 ? VERIFICA SE ACABOU A SIMULACAO  
2590 IF GAME=1 THEN CLS: GOTO 2070  
2600 RETURN  
2610 ? APAGA PARTE INFERIOR DA TELA  
2620 FOR H=1 TO 2  
2630 LOCATE 0,19+H: PRINT "  
";  
2640 NEXT H  
2650 LOCATE 0,20  
2660 RETURN
```



INTELIGÊNCIA ARTIFICIAL - I

A simulação da inteligência é um dos campos que tem recebido grande destaque na área da informática.

Simular a inteligência significa tentar fazer com que um dado computador comporte-se (tenha algumas atitudes) de modo semelhante aos seres humanos; mais precisamente, que ele “pense” como um homem.

Nos seus primórdios a ciência da inteligência artificial fez com que alguns computadores fossem capazes de simular certos jogos e, com o passar do tempo, chegamos a ter, hoje em dia, sistemas inteligentes que auxiliam o homem nas mais diversas tarefas, desde a prospecção de novos campos de petróleo até o diagnóstico de pessoas enfermas.

Também existem os sistemas especialistas que, como o próprio nome indica, possuem o conhecimento de centenas de experts em um dado assunto e muito tem contribuído para o desenvolvimento da humanidade.

Neste capítulo, e no seguinte, trataremos do setor da inteligência artificial que lida com a simulação de jogos.

Seu MSX simulará, a partir de agora, um jogo inteligente chamado LIGA 4, no qual dois participantes (você e o seu micro), cada qual na sua vez, colocam uma peça na posição disponível mais baixa numa coluna de um tabuleiro.

O tabuleiro onde ocorre a simulação tem 8 linhas e 8 colunas. Será considerado vencedor o jogador que conseguir ligar quatro de suas peças lateralmente, em qualquer direção (vertical, horizontal ou diagonal).

Observe que os programas que simulam a inteligência artificial “pensam” de maneira bastante diferente dos seres humanos.

Por exemplo, o programa da simulação LIGA 4 verifica todas as possibilidades de jogo, atribuindo valores para cada caso: atribui valores menores para possibilidades ruins e valores maiores para as melhores possibilidades.

É lógico que, ao terminar o processo de verificação das jogadas, o programa coloca a peça no local que corresponde ao maior valor, ou seja, à melhor jogada.

Lembre-se de que é o programador (ou analista) de uma dada simulação quem define o que é uma "boa jogada" e o que é uma "má jogada".

Isso significa que, até certo ponto, o programa reflete os pensamentos de seu criador; caso as idéias sejam boas e funcionais teremos uma boa simulação, caso contrário teremos uma simulação ruim.

Para jogar LIGA 4 com seu MSX digite um número entre 1 e 8 e o seu computador, automaticamente, colocará sua peça na posição disponível mais baixa da coluna cujo número você digitou.

Nas figuras abaixo as suas peças são representadas pela letra "o" e as peças do MSX são representadas pela letra "M".

Porém, na tela do seu computador as suas peças são representadas por um "X" e as peças de seu micro são representadas por um "□":

LIGA 4

DIGITE A SUA JOGADA COM UM NUMERO
ENTRE 1 e 8.

TECLE O PARA REINICIAR.

Esta é a primeira tela onde aparece o tabuleiro. O MSX permite que você inicie o jogo:

.....
.....
.....
.....
.....
.....

12345678

Sua vez...

Digitei o número 4:^{*}

.....
.....
.....
.....
....o....

12345678

Eu...

Ele (o MSX) jogou na coluna 3:

.....
.....
.....
.....
.....
..Mo....

12345678

Sua vez...

Coloquei a minha peça na coluna 6 para tentar ligar 4 peças na horizontal (colunas 4, 5, 6 e 7):

.....
.....
.....
.....
.....
..Mo.o..

12345678

Eu...

O MSX bloqueou minha tentativa colocando uma peça na coluna 5...

.....
.....
.....
.....
..MoMo..

12345678

Sua vez...

Joguei na coluna 5 para tentar formar uma diagonal:

.....
.....
.....
....O...
..MoMo..

12345678

Eu...

OPS! Ele também está formando uma linha em diagonal!!

.....
.....
.....
....Mo...
..MoMo..

12345678

Sua vez...

Jogo na 4 para continuar com a diagonal (já liguei 3 peças!):

.....
.....
.....
...o...
..Mo...
.MoMo..

12345678

Eu...

Mas... ele fez o mesmo!

Até parece que está copiando as minhas jogadas!!

.....
.....
.....
...oM...
..Mo...
.MoMo..

12345678

Sua vez...

Uso minha experiência (já joguei várias vezes...) e posicione minha peça na coluna 3:

.....
.....
.....
...oM...
..oMo...
.MoMo..

12345678

Eu...

O MSX joga na coluna 3, de modo a continuar com mais uma linha em diagonal:

```
.....  
.....  
.....  
.MoM...  
.oMo...  
.MoMo...
```

12345678

Sua vez...

Finalmente jogo na coluna 3 e:

```
.....  
.....  
.o...  
.MoM...  
.oMo...  
.MoMo...
```

12345678

OK!! VOCE GANHOU!
NAO ACREDITO QUE CONSIGA FAZER DE NOVO!

QUER TENTAR (S/N) ?

Consegui vencer com a diagonal que vai desde o fundo da coluna 6 até o topo da coluna 3.

A listagem do LIGA 4 está no próximo capítulo.



LISTAGEM DO LIGA QUATRO

Caso você queira, pode modificar o formato das suas peças e das do computador: o desenho da sua peça é definido nas linhas de 2440 a 2510 e o desenho das peças do MSX é definido nas linhas de 2570 a 2640.

Note que aos uns correspondem os pontos cheios das peças e aos zeros correspondem os espaços vazios.

Após executar esta simulação, os formatos das letras "o" e "M" estarão alterados; para fazer com que voltem ao normal digite:

SCREEN 0 ←

Recomendo muita atenção e desejo boa simulação!!

```
1000 : ****
1010 : * LIGA QUATRO *
1020 : ****
1030 :
1040 SCREEN0: KEYOFF: CLEAR &H1000: PRINT: PRINT
1050 PRINT TAB(14)"LIGA 4"
1060 PRINT
```

```
1070 PRINT " DIGITE A SUA JOGADA COM UM NUMERO ENT
RE 1 e 8.": PRINT: PRINT" TECLE 0 PARA REINICIAR."
1080 FOR F=1 TO 1000: NEXT F
1090 GOSUB 2370
1100 DIM A$(15,15): DIM B(15,2)
1110 FLAG = 0
1120 '
1130 C$ = "M": H$ = "O"
1140 FOR F=0 TO 15: FOR G=0 TO 15: A$(F,G)=CHR$(0):NEX
T G,F
1150 FOR F=5 TO 12
1160 B(F,1) = 10
1170 NEXT F
1180 FOR F=5 TO 10
1190 FOR G=5 TO 12
1200 A$(F,G) = "."
1210 NEXT G,F
1220 ' *****
1230 ' * LE SUA JOGADA *
1240 ' *****
1250 GOSUB 1450
1260 PRINT: PRINT TAB(15)"Sua vez... "
1270 A$=INKEY$: IF A$="" THEN 1270 ELSE A = VAL(A$) + 4
```

```
1280 IF A=4 THEN RUN
1290 IF A<5 OR A>12 THEN GOTO 1270
1300 L = 4
1310 IF A$(L+1,A) <> "." OR L=10 THEN GOTO 1340
1320 L = L+1
1330 GOTO 1310
1340 IF L=4 THEN GOTO 1270
1350 A$(L,A) = H$
1360 B(A,1) = B(A,1)-1
1370 GOSUB 1450
1380 GOSUB 2040
1390 GOSUB 1590
1400 GOSUB 1450
1410 GOTO 1260
1420 " ****"
1430 " * IMPRIME JOGO * "
1440 " ****"
1450 CLS: LOCATE 0,6
1460 FOR F=5 TO 10
1470 FOR G=2 TO 12
1480 PRINT TAB(15);A$(F,G);
1490 NEXT G
1500 PRINT
```

```
1510 NEXT F
1520 PRINT:PRINT TAB(15);"12345678"
1530 PRINT
1540 IF FLAG=1 THEN PRINT "GANHEI !!!": GOTO 2200
1550 RETURN
1560 " ****"
1570 " * JOGADA DO MSX * "
1580 " ****"
1590 PRINT TAB(15)"Eu..."
1600 MV = 0
1610 FOR F=5 TO 12
1620 B(F,2)=0
1630 NEXT F
1640 FOR F=5 TO 12
1650 FOR X=-1 TO 1
1660 FOR Y=-1 TO 1
1670 IF B(F,1)=0 THEN GOTO 1730
1680 IF A$(B(F,1)+X,F+Y) = CHR$(0) OR A$(B(F,1)+X,F+Y)
= "." THEN GOTO 1710
1690 IF A$(B(F,1)+X,F+Y) = H$ THEN GOSUB 1860
1700 IF A$(B(F,1)+X,F+Y) = C$ THEN GOSUB 1960
1710 NEXT Y
1720 NEXT X
```

1730 NEXT F
1740 P = 0
1750 FOR F=5 TO 12
1760 IF B(F,2)>P THEN LET P = B(F,2): N = F
1770 NEXT F
1780 A = N: L = B(N,1)
1790 A\$(B(N,1),N) = C\$
1800 B(N,1) = B(N,1)-1
1810 N = 0
1820 P = 0
1830 GOSUB 2040
1840 RETURN
1850 '
1860 MV = 2
1870 M1 = MV
1880 IF A\$(B(F,1)+(X*2),F+(Y*2)) = H\$ THEN LET MV = MV
+ 10
1890 IF A\$(B(F,1)-X,F-Y)=H\$ THEN LET MV = MV + 20
1900 IF MV <> M1+10 THEN GOTO 1920
1910 IF A\$(B(F,1)+(X*3),F+(Y*3)) = H\$ THEN LET MV = MV
+ 1000
1920 B(F,2) = B(F,2) + MV
1930 M1 = 0

1940 RETURN

1950 *

1960 MV = 2

1970 M1 = MV

1980 IF A\$(B(F,1)+(X*2),F+(Y*2)) = C\$ THEN LET MV = MV
+ 9

1990 IF A\$(B(F,1)-X,F-Y) = C\$ THEN LET MV = MV + 20

2000 IF MV <> M1+9 THEN GOTO 2020

2010 IF A\$(B(F,1)+(X*3),F+(Y*3)) = C\$ THEN LET MV = MV
+ 2000

2020 B(F,2) = B(F,2)+MV

2030 RETURN

2040 * VERIFICA SE O JOGADOR VENCEU

2050 * USA APENAS AS COMBINACOES COM A MATRIZ A\$(L,A).

2060 BEEP

2070 RESTORE 2240

2080 FOR D=1 TO 3

2090 FOR B=0 TO 3

2100 LET TT = 0

2110 FOR E=1 TO 4

2120 READ H,V: TT = TT + ASC(A\$(L+V,A+H))

2130 NEXT E

```
2140 IF TT = 4 * ASC(H$) THEN GOTO 2180
2150 IF TT = 4 * ASC(C$) THEN LET FLAG = 1: GOTO 1450
2160 NEXT B,D
2170 RETURN
2180 ' O JOGADOR VENCEU!!
2190 PRINT "OK!! VOCE GANHOU!" NAO
ACREDITO QUE CONSIGA FAZER DE NOVO!
2200 PRINT "QUER TENTAR (S/N) ?"
2210 A$=INKEY$
2220 IF A$ = "S" OR A$ = "s" THEN RUN
2230 IF A$ = "N" OR A$ = "n" THEN STOP ELSE GOTO 2210
2240 '
2250 DATA -3,0,-2,0,-1,0,0,0
2260 DATA -2,0,-1,0,0,0,1,0
2270 DATA -1,0,0,0,1,0,2,0
2280 DATA 0,0,1,0,2,0,3,0
2290 DATA -3,-3,-2,-2,-1,-1,0,0
2300 DATA -2,-2,-1,-1,0,0,1,1
2310 DATA -1,-1,0,0,1,1,2,2
2320 DATA 0,0,1,1,2,2,3,3
2330 DATA -3,3,-2,2,-1,1,0,0
2340 DATA -2,2,-1,1,0,0,1,-1
2350 DATA -1,1,0,0,1,-1,2,-2
```

```
2360 DATA 0,0,1,-1,2,-2,3,-3
2370 ' ****
2380 ' * REDEFINE LETRAS *
2390 ' ****
2400 RESTORE 2440: FOR W=0 TO 7
2410 READ A$
2420 VPOKE 2936+W,VAL("&B"+A$)
2430 NEXT W
2440 DATA 10000100
2450 DATA 01001000
2460 DATA 00110000
2470 DATA 00110000
2480 DATA 00110000
2490 DATA 01001000
2500 DATA 10000100
2510 DATA 00000000
2520 REM REDEFINE A LETRA M
2530 RESTORE 2570 : FOR W=0 TO 7
2540 READ A$
2550 VPOKE 2664+W,VAL("&B"+A$)
2560 NEXT W
2570 DATA 01111000
2580 DATA 01001000
```

2590 DATA 01001000

2600 DATA 01001000


2610 DATA 01001000

2620 DATA 01001000

2630 DATA 01111000

2640 DATA 00000000

2650 RETURN



INTELIGÊNCIA ARTIFICIAL - II

A partir deste capítulo seu MSX será capaz de disputar com você um complexo jogo de tabuleiro chamado **MSX-THELLO**.

A ação ocorre num tabuleiro de 8 x 8 casas, sendo consagrado vencedor o jogador que conseguir ocupar o maior número de casas do tabuleiro com suas peças ou eliminar todas as peças do adversário.

Existem duas regras básicas para o posicionamento das peças:

Primeira regra: cada peça colocada no tabuleiro deve tocar pelo menos uma peça do adversário.

Segunda regra: deve haver, obrigatoriamente, uma outra peça do jogador do outro lado da peça adversária com a qual sua peça atual (colocada agora no tabuleiro) faz fronteira.

Com as peças já colocadas, todas as peças do oponente que estiverem entre a peça nova e qualquer uma das peças antigas do jogador serão convertidas em peças do jogador atual.

Não importa se a linha formada entre as duas peças do jogador atual é vertical, horizontal ou diagonal ou até mesmo se ocorre nas três direções; em todos os casos, todas as peças oponentes que formem uma linha contínua entre a nova peça e as antigas serão transformadas em peças do jogador atual.

Imagine que suas peças são representadas pela letra "H" e que as peças do seu MSX são representadas por "C"s.

Se uma dada linha de peças do tabuleiro fosse

CCCCCCH

e você colocasse uma nova peça "H" na casa da esquerda, todos os "C"s se transformariam em "H"s:

sua jogada: HCCCCCCH
conseqüência: HHHHHHHH

Para posicionar sua peça nesta simulação você deve digitar dois números: primeiro o número da linha em que está a casa desejada (de 1 a 8) e, em seguida, o número da coluna (também de 1 a 8).

Esta simulação de inteligência artificial é um bom exemplo de como um programa curto pode conter um "ótimo jogador".

Lembre-se de que esta simulação envolve vários macetes e estratégias de jogo!!!

Nas telas abaixo suas peças são representadas pela letra "O" e as do MSX pela letra "X".

A primeira tela desta simulação não é muito humilde...

DESAFIO VOCE

E TODOS OS HUMANOS

A JOGAR

MSX - THE LLO

TECLE ALGO

Note que você e o MSX iniciam a simulação com duas peças cada um, sendo o micro o primeiro a jogar:

Computador: 2 1 2 3 4 5 6 7 8
Humano: 2 1
 2
 3
 4 . . . X 0 . . .
 5 . . . 0 X . . .
 6
 7
 8

MINHA JOGADA...

Ele colocou uma peça em 46 (4^a linha e 6^a coluna). Observe que a peça que estava em 45 transformou-se num "X":

Computador: 4 1 2 3 4 5 6 7 8
Humano: 1 1
 2
 3
 4 . . . X X X . .
 5 . . . 0 X . . .
 6
 7
 8

Joguei em 36; veja o que ocorreu com a peça situada em 45:

Computador: 3 1 2 3 4 5 6 7 8

Humano: 3 1

 2

 3 0 . .

 4 . . . X 0 X . .

 5 . . . 0 X . . .

 6

 7

 8

MINHA JOGADA...

O MSX jogou em 64.

Observe que no canto superior esquerdo da tela existe um placar indicando o número de peças de cada jogador:

Computador: 5 1 2 3 4 5 6 7 8

Humano: 2 1

 2

 3 0 . .

 4 . . . X 0 X . .

 5 . . . X X . . .

 6 . . . X

 7

 8

Joguei em 43:

Computador: 4 1 2 3 4 5 6 7 8

Humano: 4 1
 2
 3 0 . .
 4 . . 0 0 0 X . .
 5 . . . X X . . .
 6 . . . X
 7
 8

MINHA JOGADA...

Ih! Ele jogou em 42...

Perdi todas as peças da linha 4...

Computador: 8 1 2 3 4 5 6 7 8

Humano: 1 1
 2
 3 0 . .
 4 . X X X X X . .
 5 . . . X X . . .
 6 . . . X
 7
 8

E assim o jogo vai evoluindo, repleto de surpresas!

A listagem do MSX-THELLO está no próximo capítulo.

CAPÍTULO 10



LISTAGEM DO MSX-THELLO

Lembre-se de que esta é uma simulação complexa!!

Mas é jogando que se aprende; por isso, bom jogo e boa simulação!

```
1000 ' MSX-THELLO
1010 GOSUB 1740: REM INICIALIZACAO
1020 '
1030 LOCATE 14,20: PRINT"MINHA JOGADA . . . "
1040 S = 0: T = X: H = 0
1050 FOR A = 2 TO 9: FOR B = 2 TO 9
1060 IF A(A,B) <> 46 THEN 1210
1070 Q = 0
1080 FOR C = -1 TO 1: FOR D = -1 TO 1
1090 K = 0: F = A: G = B
1100 IF A(F+C,G+D) <> S THEN 1130
1110 K = K+1: F = F + C: G = G + D
```

```
1120 GOTO 1100
1130 IF A(F+C,G+D) <> T THEN 1150
1140 Q = Q + K
1150 NEXT D,C
1160 IF A = 2 OR A = 9 OR B = 2 OR B = 9 THEN Q = Q * 2
1170 IF A = 3 OR A = 8 OR B = 3 OR B = 8 THEN Q = Q / 2
1180 IF (A = 2 OR A = 9) AND (B = 3 OR B = 8) OR (A =
3 OR A = 8) AND (B = 2 OR B = 9) THEN Q = Q / 2
1190 IF Q < H OR (RND(-TIME) < .3 AND Q = H) THEN 1210
1200 H = Q: M = A: N = B
1210 NEXT B,A
1220 IF H = 0 AND R = 0 THEN 1670
1230 IF H = 0 THEN 1250
1240 GOSUB 1560
1250 GOSUB 1410
1260 ?
1270 LOCATE 14,20: INPUT "SUA JOGADA";R$
1280 IF R$ = "0" THEN 1300
1290 IF R$ < "11" OR R$ > "88" THEN 1270
1300 R = VAL(R$)
1310 REM DIGITE 0 PARA PASSAR A VEZ
1320 S = X: T = OX
1330 IF R = 0 THEN 1390
```

```
1340 IF R < 11 OR R > 88 THEN 1270
1350 R = R + 11
1360 M = INT(R/10)
1370 N = R - 10 * M
1380 GOSUB 1560
1390 GOSUB 1410
1400 GOTO 1020
1410 REM IMPRIME TABULEIRO
1420 LOCATE 14,20: PRINT "
1430 C = 0: H = 0
1440 LOCATE 1,1
1450 LOCATE 22,2:PRINT "1 2 3 4 5 6 7 8": PRINT: FOR B
= 2 TO 9
1460 PRINT TAB(19)B-1;
1470 FOR D = 2 TO 9
1480 PRINT TAB(20)CHR$(A(B,D))" ";
1490 IF A(B,D) = X THEN C = C + 1
1500 IF A(B,D) = OX THEN H = H + 1
1510 NEXT D: PRINT: PRINT: NEXT
1520 LOCATE 2,2: PRINT "Computador:"C
1530 LOCATE 2,4: PRINT "Humano:"H
1540 IF (C + H = 64) OR C = 0 OR H = 0 THEN 1680
1550 RETURN
```

```
1560 FOR C = -1 TO 1: FOR D = -1 TO 1
1570 F = M: G = N
1580 IF A(F+C,G+D) <> S THEN 1610
1590 F = F + C: G = G + D
1600 IF C <> 0 OR D <> 0 THEN GOTO 1580
1610 IF A(F+C,G+D) <> T THEN 1650
1620 A(F,G) = T
1630 IF M = F AND N = G THEN 1650
1640 F = F - C: G = G - D: GOTO 1620
1650 NEXT D,C
1660 RETURN
1670 GOSUB 1410
1680 LOCATE 14,20: PLAY "V15ABCD"
1690 IF C > H THEN PRINT "SOU O CAMPEAO!!"
1700 IF H > C THEN PRINT "VOCE E' O CAMPEAO!!"
1710 IF H = C THEN PRINT "EMPATAMOS!!"
1720 END
1730 * *****
1740 * INICIALIZACAO
1750 GOSUB 1900: REM TELA INICIAL
1760 *
1770 DEFINT A-Z
1780 DIM A(10,10), C(9,9), D(9,9), E(9,9), F(9,9), G(9
```

,9), H(9,9), J(9,9), K(9,9)

1790 X = 88: DX = 79

1800 FOR B = 2 TO 9: FOR C = 2 TO 9

1810 A(B,C) = 46

1820 NEXT C,B

1830 A(5,5) = X: A(6,6) = X

1840 A(6,5) = 0X: A(5,6) = 0X

1850 P = 0

1860 CLS

1870 GOSUB 1410

1880 RETURN

1890 ' *****

1900 ' TELA INICIAL

1910 KEY OFF: SCREEN 0

1920 COLOR 15,0,0

1930 WIDTH 40

1940 CLS

1950 LOCATE 13,7: PRINT "DESAFIO VOCE"

1960 LOCATE 10,9: PRINT "E TODOS OS HUMANOS"

1970 LOCATE 15,11: PRINT "A JOGAR"

1980 LOCATE 9,13: PRINT "M S X - T H E L L O"

1990 LOCATE 15,22: PRINT "TECLE ALGO"

2000 IF INKEY\$="" THEN 2000 ELSE RETURN



BOLSA DE VALORES

O seu computador trabalha somente com números, os quais podem representar desde o número de votos de cada candidato à Constituinte até o número de átomos de um dado elemento químico existente em todo o universo.

Na simulação deste capítulo vamos colocar de lado os átomos e trabalhar com dinheiro!

Você é um empresário que possui ações de cinco empresas e um capital inicial de 15.000 cruzados.

No inicio da simulação é solicitado o seu objetivo financeiro, que deve ser atingido nesta simulação:

digite um número entre 16000 e 100000 e tecle ↵ (ENTER).

Sugiro que você inicie sua carreira de empresário das ações com objetivos financeiros modestos (algo entre Cr\$ 16.000 e 17.000) e que, com o passar do tempo (e algumas simulações), aumente-os gradualmente.

Uma vez indicado o objetivo, aparecerá na tela do seu MSX a situação da Bolsa de Valores no dia:

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decremento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	1200	1000	800
------	------	------	------	-----

BANCO Cz\$ 265 VALOR TOTAL Cz\$ 15000

DESEJA VENDER ALGUMA ACAO?

Note que a cada dia você é informado sobre as chances (probabilidades) do valor de cada ação subir ou cair.

Também são impressos dados referentes à quantia que você tem no banco (BANCO) e ao valor total do seu capital (VALOR TOTAL).

Agora tecle <S> para vender ações ou <N> em caso contrário.

Caso responda <S> a próxima tela poderia ser:

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decremento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	1200	1000	800
------	------	------	------	-----

BANCO Cz\$ 265 VALOR TOTAL Cz\$ 15000

DE QUAL EMPRESA QUER VENDER?

e, depois de informar o número da empresa da qual vai vender ações:

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decremento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	1200	1000	800
------	------	------	------	-----

BANCO Cz\$ 265 VALOR TOTAL Cz\$ 15000

DE QUAL EMPRESA QUER VENDER?

QUANTAS DE 3 QUER VENDER?

? 1200

segundo-se:

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decremento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	0	1000	800
------	------	---	------	-----

BANCO Cz\$ 3253 VALOR TOTAL Cz\$ 15000

DESEJA COMPRAR ALGUMA ACAO?

Resolvi comprar ações da empresa 4:

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decrecimento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	0	1000	800
------	------	---	------	-----

BANCO Cz\$ 3253 VALOR TOTAL Cz\$ 15000

DE QUAL EMPRESA QUER COMPRAR?

QUANTAS DE 4 QUER COMPRAR?

? 1000

DIA 1 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

-7	35	-4	83	28
----	----	----	----	----

Probabilidade de decrecimento (%)

18	23	23	12	20
----	----	----	----	----

Valor atual por acao (Cz\$)

1.49	1.99	2.49	2.99	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	0	2000	800
------	------	---	------	-----

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15000

TECLE ESPACO PARA PROSSEGUIR

Você pode efetuar, no máximo, uma operação de venda e uma operação de compra de ações por dia.

A mensagem TECLE ESPAÇO PARA PROSSEGUIR indica o término de um dia de atividades na Bolsa de Valores.

Somente no próximo dia de atividades você saberá qual foi o comportamento de suas ações e de seu capital e também se atingiu o seu objetivo.

Passemos ao dia 2:

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

57	-14	29	17	-10
----	-----	----	----	-----

Probabilidade de decremento (%)

25	18	8	6	24
----	----	---	---	----

Valor atual por ação (Cz\$)

1.49	2.01	2.49	3.23	3.49
------	------	------	------	------

Número de ações que você tem

2000	1500	0	2000	800
------	------	---	------	-----

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15531

DESEJA VENDER ALGUMA AÇÃO?

Ops!! O seu capital aumentou Cz\$ 531 de um dia para outro! Nada mal para um novato, hein?!

Mostramos, a seguir, as possíveis telas de uma simulação da Bolsa de Valores com um objetivo de Cz\$ 16.000.

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

57 -14 29 17 -10

Probabilidade de decremento (%)

25 18 8 6 24

Valor atual por acao (Cz\$)

1.49 2.01 2.49 3.23 3.49

Numero de acoes que voce tem

2000 1500 0 2000 800

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15531

DE QUAL EMPRESA QUER VENDER?

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

57 -14 29 17 -10

Probabilidade de decremento (%)

25 18 8 6 24

Valor atual por acao (Cz\$)

1.49 2.01 2.49 3.23 3.49

Numero de acoes que voce tem

2000 1500 0 2000 800

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15531

DE QUAL EMPRESA QUER VENDER?

QUANTAS DE 2 QUER VENDER?

? 2000

Venderei ações da empresa 2:

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

57	-14	29	17	-10
----	-----	----	----	-----

Probabilidade de decremento (%)

25	18	8	6	24
----	----	---	---	----

Valor atual por acao (Cz\$)

1.49	2.01	2.49	3.23	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	0	2000	800
------	------	---	------	-----

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15531

DE QUAL EMPRESA QUER VENDER?

QUANTAS DE 2 QUER VENDER?

? 2000

NAO POSSUI TANTAS!!

OPS! Tentei vender o que não tinha! Agora corrigir o erro:

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

57	-14	29	17	-10
----	-----	----	----	-----

Probabilidade de decremento (%)

25	18	8	6	24
----	----	---	---	----

Valor atual por acao (Cz\$)

1.49	2.01	2.49	3.23	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	1500	0	2000	800
------	------	---	------	-----

BANCO Cz\$ 263 VALOR TOTAL Cz\$ 15531

DE QUAL EMPRESA QUER VENDER?

QUANTAS DE 2 QUER VENDER?

? 2000

NAO POSSUI TANTAS!!

? 1500

Prosseguindo

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

57 -14 29 17 -10

Probabilidade de decremento (%)

25 18 8 6 24

Valor atual por acao (Cz\$)

1.49 2.01 2.49 3.23 3.49

Numero de acoes que voce tem

2000 0 0 2000 800

BANCO Cz\$ 3283 VALOR TOTAL Cz\$ 15531

DESEJA COMPRAR ALGUMA ACAO?

Comprei ações da empresa 3, pois apresenta boa possibilidade de aumento na cotação:

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

57 -14 29 17 -10

Probabilidade de decremento (%)

25 18 8 6 24

Valor atual por acao (Cz\$)

1.49 2.01 2.49 3.23 3.49

Numero de acoes que voce tem

2000 0 0 2000 800

BANCO Cz\$ 3283 VALOR TOTAL Cz\$ 15531

DE QUAL EMPRESA QUER COMPRAR?

QUANTAS DE 3 QUER COMPRAR?

? 1000

Os valores e quantidades são atualizados:

DIA 2 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

57	-14	29	17	-10
----	-----	----	----	-----

Probabilidade de decremento (%)

25	18	8	6	24
----	----	---	---	----

Valor atual por acao (Cz\$)

1.49	2.01	2.49	3.23	3.49
------	------	------	------	------

Numero de acoes que voce tem

2000	0	1000	2000	800
------	---	------	------	-----

BANCO Cz\$ 793 VALOR TOTAL Cz\$ 15531

TECLE ESPACO PARA PROSSEGUIR

Note que as cotações das ações 1 e 4 subiram no dia 3:

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

31	32	7	93	-7
----	----	---	----	----

Probabilidade de decremento (%)

4	25	29	7	15
---	----	----	---	----

Valor atual por acao (Cz\$)

1.57	2.01	2.49	3.29	3.4
------	------	------	------	-----

Numero de acoes que voce tem

2000	0	1000	2000	800
------	---	------	------	-----

BANCO Cz\$ 793 VALOR TOTAL Cz\$ 15745

DESEJA VENDER ALGUMA ACAO?

Sim, desejo!

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

31	32	7	93	-7
----	----	---	----	----

Probabilidade de decremento (%)

4	25	29	7	15
---	----	----	---	----

Valor atual por acao (Cz\$)

1.57	2.01	2.49	3.29	3.4
------	------	------	------	-----

Numero de acoes que voce tem

2000	0	1000	2000	800
------	---	------	------	-----

BANCO Cz\$ 793 VALOR TOTAL Cz\$ 15745

DE QUAL EMPRESA QUER VENDER?

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

31	32	7	93	-7
----	----	---	----	----

Probabilidade de decremento (%)

4	25	29	7	15
---	----	----	---	----

Valor atual por acao (Cz\$)

1.57	2.01	2.49	3.29	3.4
------	------	------	------	-----

Numero de acoes que voce tem

2000	0	1000	2000	800
------	---	------	------	-----

BANCO Cz\$ 793 VALOR TOTAL Cz\$ 15745

DE QUAL EMPRESA QUER VENDER?

QUANTAS DE 5 QUER VENDER?

? 800

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

31	32	7	93	-7
----	----	---	----	----

Probabilidade de decremento (%)

4	25	29	7	15
---	----	----	---	----

Valor atual por acao (Cz\$)

1.57	2.01	2.49	3.29	3.4
------	------	------	------	-----

Numero de acoes que voce tem

2000	0	1000	2000	0
------	---	------	------	---

BANCO Cz\$ 3519 VALOR TOTAL Cz\$ 15745

DESEJA COMPRAR ALGUMA ACAO?

Vou comprar ações da empresa 4 porque é quase certeza que a cotação delas vai subir:

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1	2	3	4	5
---	---	---	---	---

Probabilidade de incremento (%)

31	32	7	93	-7
----	----	---	----	----

Probabilidade de decremento (%)

4	25	29	7	15
---	----	----	---	----

Valor atual por acao (Cz\$)

1.57	2.01	2.49	3.29	3.4
------	------	------	------	-----

Numero de acoes que voce tem

2000	0	1000	2000	0
------	---	------	------	---

BANCO Cz\$ 3519 VALOR TOTAL Cz\$ 15745

DE QUAL EMPRESA QUER COMPRAR?

QUANTAS DE 4 QUER COMPRAR?

? 1000

DIA 3 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

31 32 7 93 -7

Probabilidade de decremento (%)

4 25 29 7 15

Valor atual por acao (Cz\$)

1.57 2.01 2.49 3.29 3.4

Numero de acoes que voce tem

2000 0 1000 3000 0

BANCO Cz\$ 226 VALOR TOTAL Cz\$ 15745

TECLE ESPACO PARA PROSSEGUIR

Entramos no dia 4 com uma surpresa: o objetivo foi atingido!

DIA 4 SEU OBJETIVO E' Cz\$ 16000

Empresa numero:

1 2 3 4 5

Probabilidade de incremento (%)

19 7 53 12 48

Probabilidade de decremento (%)

6 19 14 18 30

Valor atual por acao (Cz\$)

1.57 1.96 2.49 3.59 3.4

Numero de acoes que voce tem

2000 0 1000 3000 0

BANCO Cz\$ 226 VALOR TOTAL Cz\$ 16664

OBJETIVO COMPLETO! PARABENS!!!

A listagem da Bolsa de Valores está no próximo capítulo.



LISTAGEM DA BOLSA DE VALORES

Para alterar o nível de dificuldade desta simulação substitua o número 55 na linha 230 por outro número (leia o comentário da linha 220 para maiores informações).

```
10 ' B O L S A   D E   V A L O R E S
20 '
30 '
40 DIM S!(5),N!(5),P!(5),D!(5)
50 KEY OFF
60 FOR CO=1 TO 5: READ S(CO): NEXT CO
70 DATA 1.49, 1.99, 2.49, 2.99, 3.49
80 FOR CO=1 TO 5: READ N(CO): NEXT CO
90 DATA 2000, 1500, 1200, 1000, 800
100 BB=265: TV=15000: QQ=TV: DIA=1
110 SCREEN0: PRINT:PRINT"INFORME O SEU OBJETIVO PARA E
STA SIMULA-CAO":
```

```
120 PRINT; " (de Cz$16000 a Cz$100000)"  
130 PRINT:INPUT "Objetivo: Cz$ ";OBJETIVO  
140 IF OBJETIVO<16000 THEN PRINT "MUITO BAIXO!!": GOSU  
B1330:GOTO 110  
150 IF OBJETIVO > 100000! THEN PRINT "MUITO ALTO!!": G  
OSUB 1330: GOTO 110  
160 '  
170 '*****  
180 '*      LOOP PRINCIPAL      *  
190 '*****  
200 '  
210 FOR C=1 TO 5  
220 'MODIFICA-SE O VALOR 55 DA LINHA SEGUINTE P/MUDAR  
O JOGO: 80=MUITO DIFICIL, 30=MUITO FACIL  
230 D(C)=INT (RND(9)*55) + 1  
240 P(C)=INT (RND(9)*100-D(C)) + 1  
250 NEXT C  
260 GOSUB 300  
270 GOTO 560  
280 '  
290 '*****  
300 '*      IMPRESSAO      *  
310 '*****
```

```
320 '
330 CLS
340 PRINT"-----"
350 PRINT "DIA"; DIA; "      SEU OBJETIVO E' Cz$ ";OBJE
TIVO
360 PRINT"-----"
370 PRINT "Empresa numero:"
380 FOR I=0 TO 4: PRINT TAB(2+I*6)I+1;: NEXT I: PRINT
: PRINT
390 PRINT"Probabilidade de incremento (%)"
400 FOR I=0 TO 4: PRINT TAB(2+I*6)P(I+1);: NEXT I: PR
INT:PRINT
410 PRINT "Probabilidade de decremento (%)"
420 FOR I=0 TO 4: PRINT TAB(2+I*6)D(I+1);: NEXT I: PR
INT:PRINT
430 PRINT "Valor atual por acao (Cz$)"
440 FOR I=0 TO 4: PRINT TAB(I*6+2) INT(S(I+1)*100)/100
;:NEXT I :PRINT:PRINT
450 PRINT "Numero de acoes que voce tem"
460 FOR I=0 TO 4: PRINT TAB(2+I*6) N(I+1);: NEXT I: PR
INT: PRINT
470 PRINT "BANCO Cz$" INT(BB) " VALOR TOTAL Cz$" INT(T
V)
```

```
480 PRINT"-----"
490 IF TV>OBJETIVO THEN PRINT "OBJETIVO COMPLETO! PARA
BENS!!!": STOP
500 RETURN
510 "
520 ' *****
530 '*          VENDER          *
540 ' *****
550 "
560 PRINT "DESEJA VENDER ALGUMA ACAO?"
570 A$ = INKEY$
580 IF A$<>"S" AND A$<>"N" AND A$<>"s" AND A$<>"n" THE
N GOTO 570
590 IF A$="N" OR A$="n" THEN GOTO 810
600 GOSUB 300
610 PRINT "DE QUAL EMPRESA QUER VENDER?"
620 A$ = INKEY$
630 IF A$<"1" OR A$>"5" THEN GOTO 620
640 C = VAL(A$)
650 PRINT "QUANTAS DE " C " QUER VENDER?"
660 INPUT N
670 IF N>N(C) THEN PRINT "NAO POSSUI TANTAS!!": GOTO 6
680
```

```
680 '
690 ' ****
700 '* AJUSTE DE VALORES *
710 ' ****
720 '
730 BB = BB+S(C)*N: 'SOMA VALOR NO BANCO
740 N(C) = N(C)-N: 'RETIRA ACOES VENDIDAS DO TOTAL
750 TV=0: ' COLOCA VALOR TOTAL EM ZERO
760 'DETERMINA O VALOR ATUAL
770 FOR C=1 TO 5
780 TV = TV+N(C)*S(C)
790 NEXT C
800 TV = TV+BB:' SOMA AO BALANCO DO BANCO
810 GOSUB 300
820 '
830 ' ****
840 '*          COMPRAR          *
850 ' ****
860 '
870 PRINT "DESEJA COMPRAR ALGUMA ACAO?";"
880 A$ = INKEY$
890 IF A$<>"S" AND A$<>"N" AND A$<>"s" AND A$<>"n" THE
N GOTO 880
```

```
900 IF A$=="N" OR A$=="n" THEN GOTO 1060
910 GOSUB 300
920 PRINT "DE QUAL EMPRESA QUER COMPRAR?";
930 A$ = INKEY$
940 IF A$<"1" OR A$>"5" THEN GOTO 930
950 C = VAL(A$)
960 PRINT:PRINT "QUANTAS DE " C " QUER COMPRAR?"
970 INPUT N
980 IF N*S(C)>BB THEN PRINT "NAO TEM DINHEIRO SUFICIENTE!": GOTO 970
990 '
1000 '*****
1010 ' * AJUSTE DE VALORES *
1020 '*****
1030 '
1040 BB = BB-S(C)*N
1050 N(C) = N(C)+N
1060 TV = 0
1070 FOR C=1 TO 5
1080 TV = TV+N(C)*S(C)
1090 NEXT C
1100 TV = TV+BB
1110 GOSUB 300
```

```
1120 '
1130 ' ****
1140 '* MUDA TODOS INDICADORES *
1150 ' ****
1160 '
1170 TV = 0
1180 FOR C=1 TO 5
1190 K = INT (RND(9)*100)+1
1200 IF K<P(C) THEN LET S(C)=S(C)*(1+(P(C)/1000))
1210 K = INT(RND(9)*100)+1
1220 IF K<D(C) THEN LET S(C)=S(C)/(1+(D(C)/1000))
1230 TV = TV+(S(C)*N(C))
1240 NEXT C
1250 TV = TV+BB
1260 QQ = QQ*1.005
1270 PRINT: PRINT"      TECLE ESPACO PARA PROSEGUIR"
1280 IF INKEY$ <> " " THEN 1280
1290 DIA = DIA + 1
1300 GOTO 210
1310 END
1320 ' LOOP DE TEMPO
1330 FOR JJ=1 TO 300: NEXT JJ: RETURN
```



E AGORA . . . VOAR !!

Os simuladores de voo são, provavelmente, as simulações mais conhecidas em computadores. Embora a maioria de nós nunca tenha aprendido a voar, tenho certeza de que gostaríamos de experimentar essa sensação. Um simulador de voo em um computador nos dá a oportunidade de sentir um pouco essa sensação de estar voando.

A principal dificuldade de um piloto é ter de estar atento a um número muito grande de parâmetros que mudam a cada instante e simultaneamente. Ignore qualquer um deles — como o ângulo de voo ou o nível do combustível que abaixa a cada instante — por um longo período de tempo e provavelmente irá de encontro ao solo!

Embora esta simulação não ocorra em tempo real e, portanto, você tenha a oportunidade de analisar a situação global antes de dar um comando, verá que ela é um desafio enorme.

Até a decolagem de um avião é difícil nas primeiras vezes. E, uma vez no ar, terá de lutar a cada instante para ficar aí.

E quando tentar pousar...

Sofrer acidentes no pouso é um dos perigos dos aviões reais. Você ficará muito feliz ao notar que pode tentar pousar esse simulador sem grandes riscos...

INTRODUÇÃO AOS CONTROLES

Esta simulação é a mais completa e complexa deste livro, sendo necessário um tempo maior para aprender a utilizá-la.


O tempo dispendido será amplamente recompensado tendo-se em vista que, com certeza, esta simulação trará momentos de grande lazer e emoção.

O voo é dividido em três partes ou etapas, cada qual com suas características específicas: a primeira é a decolagem, a segunda o voo em si e a terceira o pouso.

Conseguirá ser um bom piloto se dedicar-se a apenas uma das etapas de cada vez.

E mais: não desista se suas primeiras vinte tentativas de pouso terminarem em desastres. A satisfação que sentirá ao conseguir pousar perfeitamente compensará toda a frustração da etapa de aprendizagem.

Isto é o que você verá ao executar o programa pela primeira vez:



Acima, à esquerda, está o horizonte, o qual pode mover-se repentinamente de lado a lado, quando você inclinar o avião (fizer curvas ou mudar a inclinação do avião com relação ao chão).

À direita do horizonte está a bússola. O símbolo "@" indica a direção para a qual o avião está apontando. Sempre inicia-se o voo apontando para o norte.

Mais abaixo, no painel de instrumentos, pode-se observar a direção exata em que o avião está voando, na linha do ALTIMETRO, onde está escrito "0 Graus".

Os instrumentos abaixo do horizonte e da bússola informam a distância percorrida pelo avião (o raio a partir da posição inicial) e o tempo de voo.

Abaixo desses instrumentos encontram-se os indicadores da velocidade do ar, da altura em que se encontra o avião e do combustível.

A ELEVAÇÃO indica o ângulo do avião com relação ao solo; uma elevação igual a zero indica que o avião está paralelo ao solo, uma elevação de 30 graus indica que o avião está subindo com um ângulo de 30 graus em relação ao solo. Já uma elevação negativa indica que o avião está com o bico para baixo, ou seja, que ele está descendo.

Sempre que você dá um comando, ao digitar um caractere, aparece o comando correspondente na mesma linha em que ocorre a indicação da elevação. Por exemplo, ao pressionar "L" aparecerá a mensagem "ACELERAR".

Finalmente, o último instrumento do painel indica a posição do trem de aterrissagem: se está para fora ou para dentro.

Depois de alguns poucos comandos a tela pode estar assim: a velocidade do ar é 78, estamos a 210 unidades de altura, o combustível reduziu para 651 litros e o avião está formando um ângulo de 6 graus em relação ao solo:

HORIZONTE	BUSSOLA
	.N.
	..S..
	.. :
*****	O--X--E
	.. : ..

	.S.
Km. 1.8 : TEMPO 2.2 : 121	
VEL. AR: 78	
5550	
ALTIMETRO: 210 357 Graus	
55555550	
GASOLINA: 651	
5555555555555550	
ELEVACAO : 6 : ACELERAR	
>RODAS DENTRO<	

Observe agora que o horizonte está inclinado, pois o avião está virando para a direita. A direção atual é de 59 graus:

HORIZONTE	BUSSOLA
**	.N.
**
**	.. : @.
**	O--X--E
**	.. : ..
**
**	.S.

Km. 3.7 :TEMPO 3.5 : 121

VEL.AR: 72
5550

ALTIMETRO: 294 59 Grauss
5555555550

GASOLINA: 602
55555555555550

ELEVACAO : 1 :DIREITA
>RODAS DENTRO<

Dei o comando “POUSAR” e a inclinação do avião passou a ser -25 graus, sendo que o avião começou a cair (a altitude diminuiu e a velocidade aumentou!):

HORIZONTE	BUSSOLA
**	.N.
***
***	.. : @.
***	O--X--E
***	.. : ..
***
***	.S.

Km. 4.4 :TEMPO 4 : 121 1

VEL.AR: 86
5550

ALTIMETRO: 168 56 Grauss
55550

GASOLINA: 582
55555555555550

ELEVACAO : -25 :POUSAR
>RODAS DENTRO<

A situação é desesperadora!

A velocidade do ar aumentou para 102 e o avião está prestes a ir de encontro ao solo!

Rapidamente dou o comando "SUBIR":

HORIZONTE	BUSSOLA
**	.N.
*****	.. : @.
****	O--X--E
*****	.. : ..

	.S.

Km. 4.8 : TEMPO 4.3 : 121

VEL.AR: 102

555550

ALTIMETRO: 3 56 Grauss

0

GASOLINA: 569

55555555555550

ELEVACAO : -23 : SUBIR

>RODAS DENTRO<

Mas o resultado é inevitável...:

A última tela da simulação é bastante significativa:

HORIZONTE	BUSSOLA
* *C R** A **	.N.
** A ** S* H*	...:..
R** A ** S*	.. : @.
*C R** A ** S	O--X--E
	.. : ..
	...:..
	.S.
Km. 4.9 :TEMPO 4.4 : 121	
VEL.AR: 106	
555550	
ALTIMETRO: 0 55 Graus	
0	
GASOLINA: 565	
55555555555550	
ELEVACAO :-20 :DECOLAR	

No capítulo seguinte você será introduzido aos CONTROLES DE VÔO.



INTRODUÇÃO AOS CONTROLES DE VÔO

COMO LEVANTAR VÔO:

Para levantar vôo é necessário que o avião esteja com uma velocidade entre 45 e 60, e o grau de elevação deve ser maior que 10 (o ideal é algo próximo de 15 graus).

As telas a seguir têm a função de auxiliá-lo nas suas primeiras decolagens.

A tela abaixo é a tela inicial da simulação:

HORIZONTE	:	BUSSOLA
		.N.
		..E..
		--:--
*****	:	O---X---E
		--:--
	
		.S.
Km. 0	:	TEMPO .1 : 249
VEL.AR:	0	
0		
ALTIMETRO:	0	0 Graus
0		
GASOLINA:	750	
555555555555555550		
ELEVACAO :	0	:55555555555555
>RODAS P/FORAK		

Vou acelerar o avião (tecla L) para atingir a velocidade de decolagem:

HORIZONTE	BUSSOLA
	.N.
	.S..
	-- : --
*****	O---X---E
	-- : --

	.S.
Km. 0 : TEMPO .2 : 249	
VEL.AR: 7	
0	
ALTIMETRO: 0	0 Graus
0	
GASOLINA: 745	
99999999999999990	
ELEVACAO : 0 : ACELERAR	
>RODAS P/FORAK	

Após alguns comandos de "ACELERAR" é esta a situação geral:

HORIZONTE	BUSSOLA
	.N.
	.S..
	-- : --
*****	O---X---E
	-- : --

	.S.
Km. .2 : TEMPO .6 : 249	
VEL.AR: 35	
50	
ALTIMETRO: 0	0 Graus
0	
GASOLINA: 725	
99999999999999990	
ELEVACAO : 0 : ACELERAR	
>RODAS P/FORAK	

Embora a velocidade do avião não seja a ideal, vou começar a incliná-lo (tecla Q) em relação ao solo:

```

HORIZONTE : BUSSOLA
| .N.
| ..E..
| ..:..
| ***** O---X---E
| ..:..
| ..:..
| .S.

Km. .3 : TEMPO .7 : 249

VEL.AR: 32
50
ALTIMETRO: 0      0 Graus
0
GASOLINA: 720
666666666666666660

ELEVACAO : 5 :DECOLAR
>RODAS P/FORAK

```

Note que a velocidade do avião diminui devido à sua inclinação.

Dou mais um comando “DECOLAR” (tecla Q) para atingir a inclinação de decolagem (a velocidade continua caindo...):

```

HORIZONTE : BUSSOLA
| .N.
| ..E..
| ..:..
| ***** O---X---E
| ..:..
| ..:..
| .S.

Km. .4 : TEMPO .9 : 249

VEL.AR: 24
50
ALTIMETRO: 0      0 Graus
0
GASOLINA: 711
666666666666666660

ELEVACAO : 15 :DECOLAR
>RODAS P/FORAK

```

Agora devo acelerar o avião para atingir uma boa velocidade de decolagem:

HORIZONTE	BUSSOLA
	.N.
	..@..
*****	..:..
	O---X---E
	..:..
	...:..
	.S.
Km. .7 :TEMPO 1.3 : 249	
VEL.AR: 40	
550	
ALTIMETRO: 0	0 Graus
0	
GASOLINA: 692	
555555555555555550	
ELEVACAO : 15 :ACELERAR	
>RODAS P/FORAK	

Como a velocidade não é suficiente, acelero mais o avião, que finalmente levanta vôo (observe o altímetro):

HORIZONTE	BUSSOLA
	.N.
	..@..
*****	..:..
	O---X---E
	..:..
	...:..
	.S.
Km. .9 :TEMPO 1.5 : 249	
VEL.AR: 48	
550	
ALTIMETRO: 25	0 Graus
0	
GASOLINA: 683	
555555555555555550	
ELEVACAO : 15 :ACELERAR	
>RODAS P/FORAK	

Para fazer um bom vôo e economizar combustível recolho o trem de aterrissagem (tecla 1). Note que a altitude do avião aumenta, devido à redução do atrito de suas rodas com o ar:

HORIZONTE	BUSSOLA
	.N.
	--S..
*****	--:--
	O---X---E
	--:--

	.S.
Km. 1 :TEMPO 1.6 : 249	
VEL.AR: 44	
550	
ALTIMETRO: 50 359 Graus	
50	
GASOLINA: 678	
5555555555555550	
ELEVACAO : 15 :5555555555555	
>RODAS DENTRO<	

COMO POUSAR:

Para pousar, a velocidade do avião deve ser inferior a 20, o trem de pouso deve estar para fora, a direção de vôo não deve variar mais de 12 graus daquela indicada abaixo da bússola, e o altímetro deve marcar menos que 15.

COMANDOS EXISTENTES:

- “L” : usado para acelerar o avião;
- “.” : usado para desacelerar o avião;
- “Q” : usado para aumentar a elevação (inclinação);
- “A” : usado para diminuir a elevação do avião;
- “Z” : usado para fazer curvas à esquerda;
- “M” : usado para fazer curvas à direita;
- “1” : usado para mudar o trem de pouso de dentro para fora ou vice-versa.

A listagem completa do SIMULADOR DE VÔO está no capítulo seguinte.



LISTAGEM DO SIMULADOR DE VÔO

Para modificar o nível de dificuldade do pouso, altere as linhas 1560 e 1570.

Para obter uma simulação em TEMPO REAL, modifique a linha 1270

de 1270 IF X\$ = " " THEN GOTO 1230

para 1270 REM

Finalizando, esta simulação só aceita os comandos com letras maiúsculas; tenha o cuidado de colocar o teclado do seu MSX na modalidade de maiúsculas com o uso da tecla <CAPS LOCK>, antes de iniciar o seu vôo.

1000 : *****

1010 : *

*

1020 : *S I M U L A D O R*

1030 : * D E *

1040 : * V O O *

1050 : *

1060 : *****

1070 :

```
1080 CLEAR &H1000: RPT = 0
1090 LD = INT(RND(-TIME)*360)+1
1100 DIM E$(1000): REM GUARDA REGISTRO DE VOO
1110 DIM C$(7), A$(7): REM GUARDAM O HORIZONTE E BUSSO
LA
1120 SCREEN1:CLS
1130 " ****"
1140 " * INICIALIZACAO * "
1150 " ****"
1160 GOSUB 3520: REM INICIALIZA VARIAVEIS
1170 IF CRASH=0 THEN GOSUB 1930: REM HORIZONTE E BUSSO
LA
1180 GOSUB 1600: REM IMPRESSAO DA TELA
1190 IF CRASH=1 THEN END: REM FIM SIMULACAO
1200 IF TIE=1 AND TFLAG=1 THEN PRINT "UM POUSO PERFEITO!! OTIMO!!": FOR ZX=250 TO 0 STEP -2: SOUND 0,ZX: SOUND 1,1: SOUND 7,254: SOUND 8,16: SOUND 12,50: SOUND 13,1: NEXT ZX: END: REM FIM SIMULACAO COM POUSO OTIMO
1210 IF TIE=1 AND TFLAG=0 THEN PRINT "O TREM DE POUSO ESTA' LEVANTADO.": GOSUB 2930: GOTO 11B0
1220 T=VAIRE: MOT=0
1230 X$=INKEY$
1240 IF X$="R" THEN RPT=1: GOTO 1160
```

```
1250 IF RPT=1 AND E$(RELOJ+1)="" THEN RPT=0: GOTO 1230
1260 IF RPT=1 THEN X$=E$(RELOJ+1)
1270 IF X$="" THEN GOTO 1230
1280 IF RELOJ<999 THEN E$(RELOJ+1)=X$
1290 IF TA=1 THEN ELEVA=INT(ELEVA+RND(-TIME)*2-RND(-TIME)*3)
1300 IF VAIRE<3 THEN 1380
1310 IF X$="Q" THEN ELEVA = ELEVA+5: EFLAG=5: IF ELEVA>60 THEN MOT=1
1320 IF X$="A" THEN ELEVA=ELEVA-5: EFLAG=-5: IF ELEVA<-70 THEN MOT= -1
1330 IF MOT<>0 THEN GOSUB 2780
1340 IF ALTI<1 THEN 1380: REM PARA EVITAR VOLTAS NA TERRA
1350 IF X$="Z" THEN AA=AA-.5: ANG=ANG-6: IF AA<-3 THEN AA= -3
1360 IF X$="M" THEN AA=AA+.5: ANG=ANG+6: IF AA>3 THEN AA=3
1370 ANG = INT(ANG+RND(-TIME)*2-RND(-TIME)*2)
1380 IF X$="L" THEN VAIRE=VAIRE+8.5
1390 IF X$=.." THEN VAIRE=VAIRE-7
1400 VAIRE = VAIRE - ELEVA/5
1410 IF TFLAG=1 THEN VAIRE = VAIRE - 1.5: FUEL = FUEL -.5
```

```
1420 IF VAIRE<0 THEN VAIRE = 0
1430 IF VAIRE>400 THEN VAIRE = 400
1440 IF X$=="1" AND TFLAG=0 THEN TFLAG=1: GOTO 1460
1450 IF X$=="1" AND TFLAG=1 THEN TFLAG=0
1460 FUEL = FUEL - (ABS(T-VAIRE)/10) - 3.75
1470 IF FUEL<1 THEN GOSUB 2930
1480 IF TA=1 THEN 1510
1490 IF ELEVA>10 AND VAIRE>45 AND VAIRE<60 AND TFLAG=1
THEN TA=1
1500 IF TA=0 THEN ALTI=0: GOTO 1540
1510 IF TIE=0 AND VAIRE<30 THEN ELEVA = ELEVA - 5: ALT
I = 9*ALTI/10
1520 ALTI = ALTI + INT(((ELEVA+.1)*VAIRE)+EFLAG*VAIRE/1
000)/B0
1530 IF ALTI<300 AND TA=1 THEN ALTI = ALTI + VAIRE/30
+ ELEVA
1540 IF ALTI<0 THEN GOSUB 2930: REM CRASH!!!
1550 ' MODIFIQUE AS DUAS LINHAS ABAIXO          PARA MOD
IFICAR O NIVEL DE DIFI-          CULDADE NO POUZO
1560 IF ALTI>15 AND VAIRE>20 OR TA=0 THEN GOTO 1170
1570 IF ABS(ANG-LD)<13 OR ABS(ANG+360-LD)<13 THEN TIE
= 1
1580 GOTO 1170
```

```
1590 * ****
1600 *      IMPRESSAO      *
1610 * ****
1620 LOCATE 4,0: PRINT STRING$(30,45): LOCATE 4,10: PR
INT STRING$(30,45): LOCATE 4,19: PRINT STRING$(30,45):
LOCATE 4,22: PRINT STRING$(30,45)
1630 FOR T=0 TO 22: LOCATE 3,T: PRINT CHR$(124): LOCAT
E 34,T: PRINT CHR$(124): NEXT T: FOR T=1 TO 9: LOCATE
22,T: PRINT CHR$(124): NEXT T
1640 LOCATE 9,1: PRINT "HORIZONTE": LOCATE 25,1: PRINT
"BUSSOLA"
1650 EV = INT(ELEVA/10)
1660 IF EV>2 THEN EV=2
1670 IF EV<-2 THEN EV= -2
1680 IF EV<>0 AND TA=1 AND CRASH=1 THEN GOSUB 3080
1690 LOCATE 4,3
1700 FOR J=1 TO 7
1710 LOCATE 6,J+2: PRINT A$(J): LOCATE 25,J+2: PRINT C
$(J)
1720 A$(J) = ""
1730 NEXT J
1740 DIST = DIST + ABS((COS(ELEVA))*VAIRE)/360
1750 RELOJ = RELOJ+1
```

```
1760 LOCATE 5,11: PRINT "Km."; INT(DIST*10)/10: LOCATE  
14,11: PRINT CHR$(58); "TEMPO"; INT(RELOJ)/10; CHR$(58); LD  
1770 LOCATE 5,13  
  
1780 PRINT "VEL.AR: "; INT(VAIRE)  
  
1790 LOCATE 5,14: PRINT LEFT$(Q$, INT(VAIRE/20)); CHR$(20  
7); STRING$(25-INT((VAIRE/20)), 32)  
  
1800 LOCATE 5,15: PRINT "ALTIMETRO: "; INT(ALTI);  
1810 IF ANG<0 THEN PRINT TAB(22); 360+ANG; "Graus"  
1820 IF ANG>=0 THEN PRINT TAB(22); ANG; "Graus"  
1830 MR = INT(ALTI/30): IF MR>20 THEN MR=20  
  
1840 LOCATE 5,16: PRINT LEFT$(Q$, MR); CHR$(207); STRING$  
(25-MR, 32)  
  
1850 LOCATE 5,17: PRINT "GASOLINA: "; INT(FUEL)  
  
1860 LOCATE 5,18: PRINT LEFT$(Q$, INT(FUEL/40)); CHR$(20  
7); STRING$(25-INT(FUEL/40), 32)  
  
1870 PRINT  
  
1880 LOCATE 5,20: PRINT "ELEVACAO : "; ELEVA; CHR$(58);:  
GOSUB 3380: PRINT U$  
  
1890 IF TFLAG=1 THEN LOCATE 5,21: PRINT ">RODAS P/FORA  
<"  
  
1900 IF TFLAG=0 THEN LOCATE 5,21: PRINT ">RODAS DENTRO  
<"  
  
1910 RETURN
```

```
1920 ' *****
1930 ' * HORIZ.E BUSSOLA *
1940 ' *****
1950 WW=ABS(INT(AA+.5))
1960 IF WW=3 THEN GOSUB 2110
1970 IF WW=2 THEN GOSUB 2200
1980 IF WW=1 THEN GOSUB 2290
1990 IF WW=0 THEN GOSUB 2380
2000 ' AS LINHAS ABAIXO ENDIREITAM      AS ASAS
      LENTAMENTE
2010 IF AA>0 THEN AA = AA-.2
2020 IF AA<0 THEN AA = AA+.2
2030 IF AA>.2 THEN 2480
2040 FOR Z=1 TO 7
2050 M$(B-Z) = A$(Z)
2060 NEXT Z
2070 FOR Z=1 TO 7
2080 A$(Z) = M$(Z)
2090 NEXT Z
2100 GOTO 2480
2110 ' AA VALE 3 OU -3
2120 A$(1) = "          ** "
2130 A$(2) = "          ** "
```

2140 A\$(3) = " ** "

2150 A\$(4) = " ** "

2160 A\$(5) = " ** "

2170 A\$(6) = " ** "

2180 A\$(7) = " ** "

2190 RETURN

2200 > AA VALE 2 OU -2

2210 A\$(1) = " "

2220 A\$(2) = " ***"

2230 A\$(3) = " *** "

2240 A\$(4) = " *** "

2250 A\$(5) = " *** "

2260 A\$(6) = " *** "

2270 A\$(7) = " "

2280 RETURN

2290 > AA VALE 1 OU -1

2300 A\$(1) = " "

2310 A\$(1) = " "

2320 A\$(3) = " *****"

2330 A\$(4) = " **** "

2340 A\$(5) = " ***** "

2350 A\$(6) = " "

2360 A\$(7) = " "

```
2370 RETURN
2380 ? AA VALE ZERO
2390 A$(1) = " "
2400 A$(2) = " "
2410 A$(3) = " "
2420 A$(4) = "*****"
2430 A$(5) = " "
2440 A$(6) = " "
2450 A$(7) = " "
2460 RETURN
2470 ? *****
2480 ? *ACERTO DE BUSSOLA*
2490 ? *****
2500 F2 = ANG-F1
2510 IF F2<0 THEN FA = INT((F2+375)/30)
2520 IF F2>=0 THEN FA = INT((F2+15)/30)
2530 IF FA=12 THEN FA=0
2540 C$(1) = ".N. "
2550 IF FA=11 THEN C$(2) = ".@...": GOTO 2590
2560 IF FA=0 THEN C$(2) = "...@..": GOTO 2590
2570 IF FA=1 THEN C$(2) = "...@.": GOTO 2590
2580 C$(2) = "...":
2590 IF FA=10 THEN C$(3) = ".@ : ..": GOTO 2620
```

```
2600 IF FA=2 THEN C$(3) = "... : @.": GOTO 2620
2610 C$(3) = "... : .."
2620 IF FA=9 THEN C$(4) = "0@-X--E": GOTO 2650
2630 IF FA=3 THEN C$(4) = "0--X-@E": GOTO 2650
2640 C$(4) = "0--X--E"
2650 IF FA=8 THEN C$(5) = ".@ : ..": GOTO 2680
2660 IF FA=4 THEN C$(5) = "... : @.": GOTO 2680
2670 C$(5) = "... : .."
2680 IF FA=7 THEN C$(6) = " .@... ": GOTO 2720
2690 IF FA=6 THEN C$(6) = " ..@.. ": GOTO 2720
2700 IF FA=5 THEN C$(6) = " ...@. ": GOTO 2720
2710 C$(6) = " ...:.. "
2720 C$(7) = " .S. "
2730 IF ANG>360 THEN ANG=ANG-360
2740 F2 = ANG
2750 IF W>0 THEN W = W-.4
2760 IF W<0 THEN W = W+.4
2770 RETURN
2780 REM
2790 IF MOT= -1 THEN 2860
2800 FOR J=1 TO 10
2810 LOCATE J,25: PRINT "P A R O U ..."
2820 FOR I=1 TO 100: NEXT I: PRINT
```

```
2830 NEXT J
2840 VAIRE = VAIRE/4
2850 RETURN
2860 FOR J=1 TO 10
2870 PRINT TAB(J); "CAINDO!!!"
2880 PRINT TAB(21-J); "PARE!!!"
2890 NEXT J
2900 ALTI = 4*ALTI/5
2910 CLS: RETURN
2920 ? ****
2930 ? *      CRASH!      *
2940 ? ****
2950 CRASH=1: ALTI=0
2960 BEEP: BEEP
2970 M$="* *C R** A ** S* H* !!*": REM 25 LETRAS DE C
DMPRIMENTO
2980 LOCATE 1,25: FOR J=1 TO 20
2990 PRINT TAB(J+5); "CRASH! "
3000 PRINT TAB(21-J); "CRASH! "
3010 NEXT J
3020 FOR J=1 TO 7
3030 G = INT(RND(1)*11)+1
3040 A$(J) = MID$(M$,G,14)
```

```
3050 NEXT J
3060 CLS: RETURN
3070 ' *****
3080 ' *AJUSTA HORIZONTE*
3090 ' *****
3100 G$ = STRING$(14,32)
3110 ON EV+3 GOSUB 3130,3190,3240,3250,3310
3120 RETURN
3130 ' EV VALE -2
3140 FOR J=1 TO 4
3150 A$(J) = A$(J+3)
3160 NEXT J
3170 A$(5)=G$: A$(6)=G$: A$(7)=G$
3180 RETURN
3190 ' EV VALE -1
3200 FOR J=1 TO 6
3210 A$(J)=A$(J+1)
3220 NEXT J
3230 A$(J)=G$
3240 RETURN
3250 ' EV VALE 1
3260 FOR J=6 TO 1 STEP -1
3270 A$(J+1) = A$(J)
```

```
3280 NEXT J
3290 A$(1)=G$
3300 RETURN
3310 ? EV VALE 2
3320 FOR J=4 TO 1 STEP -1
3330 A$(J+3) = A$(J)
3340 NEXT J
3350 A$(1)=G$: A$(2)=G$: A$(3)=G$
3360 RETURN
3370 ? *****
3380 ? *ORDENS DIGITADAS*
3390 ? *****
3400 U$ = STRING$(12,219)
3410 IF X$ = "L" THEN U$= "ACELERAR"
3420 IF X$ ="." THEN U$= "FREAR"
3430 IF X$ = "Q" AND ALTI>0 THEN U$= "SUBIR"
3440 IF X$ = "Q" AND ALTI=0 THEN U$= "DECOLAR"
3450 IF X$ = "A" THEN U$= "POUSAR"
3460 IF X$ = "Z" THEN U$= "ESQUERDA"
3470 IF X$ = "M" THEN U$= "DIREITA"
3480 G$=STRING$(10,32)
3490 IF LEN(U$)<12 THEN U$=U$+RIGHT$(G$,12-LEN(U$))
3500 RETURN
```

3510 * *****
3520 * INICIALIZACAO *
3530 * *****
3540 KEYOFF: SCREEN0: COLOR15,0,0
3550 POKE -853,255
3560 Q\$ = STRING\$(20,219)
3570 TFLAG=1: REM TREM DE POUSO: 1 = FORA,
0 = DENTRO
3580 EFLAG=0: REM ANGULO SUBIDA
3590 ANG=0: TA=0: TIE=0
3600 VAIRES=0: REM VELOCIDADE DO AR
3610 DIST=0: REM DISTANCIA PERCORRIDA
3620 ALTI=0: REM ALTITUDE
3630 ELEVA=0: REM ANGULO DE ELEVACAO
3640 AA=0: REM ANGULO DAS ASAS; USADO PARA IMPRI
MIR O HORIZONTE
3650 FUEL=750: CRASH=0: F2=0: F1=0:REM USADOS NO
DESENHO DA BUSSOLA
3660 RELOJ=0: REM MARCA TEMPO
3670 X\$=""
3680 GOSUB 2390
3690 RETURN

Composição e Arte:
JAG Composições e Artes Gráficas Ltda

Fotolito do Miolo
Binhos Fotolito S/C Ltda.

Impresso na **Prol** editora gráfica ltda.
03043 Rua Martim Burchard, 246
Brás - São Paulo - SP
Fone: (011) 270-4388 (PABX)
com filmes fornecidos pelo Editor.

OUTROS LIVROS NA ÁREA:

- Burd / Moreira — MSX — Guia do Operador
Burd / Moreira — MSX — Jogos 3 volumes
Casari — MSX — Com Disk Drive
Hoffman — MSX — Guia do Usuário

0-07-450054-6

